

情報科学概論 3 回目

今回の目的

- グラフの作成と可視化を行う.

1 可視化ソフトウェア

gnuplot, mathematica, Matlab, Maxima など

2 gnuplot

簡単に、2D, 3D のグラフを作成できるフリーのプログラム

- 参考になるホームページ (以下の他にもいろいろある)

- 本家 <http://www.gnuplot.info>
- gnuplot の初歩 <http://graph.pc-physics.com>
- コマンド集 <http://www.gnuplot-cmd.com>

- 起動方法

ターミナルで下記のように

```
% gnuplot
```

と打ち込む (%はプロンプトなのでタイプする必要は無い). すると

```
gnuplot>
```

と入力待ち状態になる.

- 終了方法

```
gnuplot> exit
```

または

```
gnuplot> quit
```

- 計算機能

- 実数

```
gnuplot> print 1+2*3
```

```
7
```

```
gnuplot> print sin(pi/4.0) #関数の引数は小括弧 ( ) で囲う. 中括弧や大括弧は別の意味がある.
```

```
0.707106781186547
```

```
gnuplot> print 2**10 # べき乗
```

```
1024
```

- 複素数

```
gnuplot> print {3,2}**2
```

```
{5.0, 12.0}
```

```
gnuplot> print abs({2,5}) # 絶対値
```

```
5.3851648071345
```

なお, #以下はコメント文なので何も処理をしない (タイプする必要は無い).

- 2次元グラフの表示

まず

```
gnuplot> set term qt
```

として、以下の関数を実行していく (set term はグラフの出力ソフト (ターミナル) を選択するコマンド。よく使われるターミナルとして X11 や AquaTerm があるが、ここではサテライトラボの PC に合わせて Qt を指定している。どのターミナルでも概ね同じ結果が得られるので自宅の PC に gnuplot をインストールする場合は何を使っても良い。通常ターミナルは gnuplot を立ち上げた時に自動で指定される¹⁾。

- sin 関数の表示

```
gnuplot> plot sin(x)
```

- x の範囲を $-2\pi \sim 2\pi$ とする。

```
gnuplot> plot [-2*pi:2*pi] sin(x)
```

- x の範囲を $-2\pi \sim 2\pi$ 、y の範囲を $-1.5 \sim 1.5$ とする。

```
gnuplot> plot [-2*pi:2*pi] [-1.5:1.5] sin(x)
```

- 複数表示

```
gnuplot> plot [-2*pi:2*pi] [-1.5:1.5] sin(x), cos(x), tan(x), x**0.5
```

- ラベルの表示方法

```
gnuplot> set xlabel "time"
```

```
gnuplot> set ylabel "intensity"
```

```
gnuplot> plot [-1:3] [0:3] exp(-x) title "observation"
```

- 範囲の指定は plot コマンドは別個に行うこともできる。x の範囲を $-1 \sim 3$ とする場合は

```
gnuplot> set xrange [-1:3]
```

- フォントを変えたい場合は、例えば下記のようにすると、Arial の 16 ポイントフォントが適応される

```
gnuplot> set terminal qt font "Arial,16"
```

- **重要：長いコマンドは講義資料をコピー&ペースしても良いが、`'` や `"` 等の記号は見かけは似ているが異なるフォントで PDF 表示されていることが多々有る。初めのうちはコマンドを覚える意味でも自分でキーボードから打つことを勧める。**

- 3次元グラフの表示

画面をクリック → ドラッグすると図が回転します (出力ターミナルが qt もしくは x11(X Window System) の場合のみなので注意)。

```
gnuplot> splot x**2+y**2
```

```
gnuplot> set isosample 40,40 #網の細かさ
```

```
gnuplot> set hidden3d #表裏を区別
```

```
gnuplot> splot cos((x**2+y**2)/5.0)/(x**2+y**2+5.0)
```

- 媒介変数表示

```
gnuplot> set parametric
```

```
dummy variable is t for curves, u/v for surfaces
```

```
gnuplot> set samples 1024
```

```
gnuplot> plot sin(5.0*t), cos(3.0*t+0.5*pi)
```

```
gnuplot> unset parametric
```

set したパラメータは unset するとデフォルトに戻る。

- 関数の定義

1 変数の場合

¹A250 の PC 利用者への注意：2019 年の実習機の入替わり前に gnuplot を利用したことがある場合は、先代の実習機で用いられていた別のターミナル (X11 もしくは aquaterm) がデフォルトに設定されている場合がある。もし qt をデフォルトに変更したい場合は、自分のホーム領域直下にある .gnuplot ファイルをテキストエディタで開き、ターミナルを指定している部分 set terminal x11 を削除するか、#set terminal x11 としてコメントアウトすれば良い。

```

gnuplot> f(x)=sin(x)
gnuplot> g(x)=x**2
gnuplot> plot f(x)*g(x)
gnuplot> print f(2)*g(3)
8.18367684143113
gnuplot> show functions
User-Defined Functions:
f(x)=sin(x)
g(x)=x**2

```

2 変数の場合

```

gnuplot> h(x,y) = exp(x)*(sin(y)+1.0)
gnuplot> set log y # 縦軸のみを対数で
gnuplot> plot h(x,0.0) # exp(x) の 2 次元プロット
gnuplot> unset log # 対数を止める
gnuplot> splot [0:][0:][:] h(x,y) # 3 次元プロット (x,y 軸は正のみ;z 軸は任意)

```

- ヘルプ

```

gnuplot> help plot
gnuplot> help expressions functions
gnuplot> help commands

```

- 出力結果をファイルに保存する.

1. 画面で出力が正確か確かめる.
2. 出力デバイスを指定する (`set terminal *`).
3. 出力デバイスのオプションを設定する
4. `set out "filename"` でファイル名の出力したいファイル名を設定.
5. コマンドを実行する.
6. `set output` でファイルを閉じる.
7. `set term qt` で画面に表示するように変更する.

- terminal の種類

png, jpeg, postscript, pdf, qt, x11, postscript など

- 例: 3D 表示の例を PNG フォーマットで `test.png` という名前で出力する

```

gnuplot> set isosample 40,40
gnuplot> set hidden3d
gnuplot> splot cos((x**2+y**2)/5.0)/(x**2+y**2+5.0)
gnuplot> set term png
Terminal type set to 'png'
Options are 'nocrop medium '
gnuplot> set out "test.png"
gnuplot> splot cos((x**2+y**2)/5.0)/(x**2+y**2+5.0)
gnuplot> set output
gnuplot> set term qt
Terminal type set to 'qt'
Options are '0'
gnuplot> !open test.png

```

- 最後の行の `!` はシェルのコマンドを呼び出す記号. `!ls` など.

- 複雑な例

```

gnuplot> set dummy u,v
gnuplot> set key bmargin center horizontal Right noreverse enhanced autotitles nobox
gnuplot> set parametric
dummy variable is t for curves, u/v for surfaces
gnuplot> set view 50, 30, 1, 1
gnuplot> set isosamples 50, 20
gnuplot> set hidden3d offset 1 trianglepattern 3 undefined 1 altdiagonal bentover
gnuplot> set ticslevel 0
gnuplot> set title "Interlocking Tori"
gnuplot> set urange [ -pi : pi ] noreverse nowriteback
gnuplot> set vrange [ -pi : pi ] noreverse nowriteback
gnuplot> set zrange [ * : * ] noreverse nowriteback
gnuplot> splot cos(u)+.5*cos(u)*cos(v),sin(u)+.5*sin(u)*cos(v),.5*sin(v)
with lines,1+ cos(u)+.5*cos(u)*cos(v),.5*sin(v),sin(u)+.5*sin(u)
*cos(v) with lines

```

- bmargin center horizontal Right はグラフ領域の下之余白中央に水平に右揃で凡例を書けという指示
- set view は三次元グラフの視点を調整するコマンド (set view x 軸周りの回転角, z 軸周りの回転角, xy 軸の拡大率, z 軸の拡大率で指定)
- ticslevel は三次元グラフの xy 平面の位置を調整するコマンド。デフォルトは 0.5 で、z 軸の原点を xy 平面と一致させるには 0 を指定。
- hidden3d のオプション offset ? は裏面の線種の指定、trianglepattern ? はポリゴンのつなげ方の指定、その他は定義されていないデータ点の処理法等の例外処理オプション
- (no)reverse: 軸の範囲を反転させるオプション
- (no)writeback: グラフを描画したときの軸の範囲を記憶させるオプション。最後に保存した軸の範囲は例えば set xrange restore とすると復元できる。

この例では複雑な調整も可能であることを示すためにわざとオプションを多用している。
あるコマンドのオプションを詳しく知りたい場合は、help コマンドが利用できる (例: help hidden3d、最近
はヘルプを解説するよりもネット検索の方が便利な場合が多い)。
長いコマンドをいちいち打つのが面倒な場合は、以下のアニメーションでやるように、複数のコマンドを 1
つのファイルにまとめて処理することもできる。こうすれば最初に 1 度ファイルを作成しておけば、以降コ
マンドを打つ必要が無くなる。

- アニメーション

以下のようなファイルを作成し、ファイル名を'sample1.gp' などとする。

```

# for sin curve
if (exist("i")==0 || i<0) i=0 #変数の初期化 (exist("x") は x が定義されていなければ 0 を
                                #返す関数、|| はもしくはという論理記号)

set parametric
set pointsize 3
# set multiplot
plot [0.01*i:0.01*i+0.01][0:10][-1:1]t,sin(t) with points pointtype 7
plot [0:0.01*i][0:10][-1:1]t,sin(t) with line
i = i+1
pause 0.01
if (i < 1000) reread
i = 0 #再稼働のときの為に終了前に i を 0 にしておく

```

そして、gnuplot を立ち上げ、

```
gnuplot> load "sample1.gp"
```

とタイプすると sin 関数のアニメーションが表示される (Ctrl-c で終了)。
コマンドが長い場合は、複数のコマンドを上記のように 1 つのファイルにまとめ、そのファイルを読み込ま
せる (load する) ことにより処理することができる。なお、#に続く行はコメント行であり、特に処理を何も
しない。上記で# set multiplot の部分は環境に合わせコメントイン (#を消去) しても良い (multiplot は

本来は複数のグラフを1つのターミナルに出力する機能だが、今の場合これを使うと環境によっては描画がスムーズになる).

3 自宅のPCでgnuplotを使うには？

- gnuplot はフリーソフトなので各自でインストールすれば利用することができる. ただし, インストール方法は OS とそのバージョンの詳細に依存するので, この授業では下記のアドレスから利用可能なオンラインサービスの利用を推奨する.
- <http://gnuplot.respawned.com>
 - 上記 web ページを開くと, 「Plot script」の項に gnuplot コマンドを入力することができる.
 - Plot script には初めから雛形となるスクリプトが記入されており, 例えば `set terminal` から始まる2行目をいじれば画像サイズやフォントサイズを変更できる. 雛形では6行目がラベルの位置指定, 7行目, 8行目が軸の説明, 9行目がタイトル, 10行目に「Data」の項に入力されているデータをプロットする具体的指示がなされている.
 - 結果は「Result」の項に表示される. 通常の gnuplot とは異なり, ターミナルを選択することはできず, 表示された図を右クリックして png 形式で保存することになる (実習の課題は png で保存できれば提出できるので安心して良い).
- もちろん, 各自でインストールできれば上記のオンラインサービスを利用する必要はない. ただし, 対応が非常に複雑かつ長い時間がかかるので, 個別のインストール相談を受け付けることはできない. インストールする知識がなければオンラインサービスを利用すること.